基于 TSSP 虚时演化求解 BEC 中的 GP 方程

赵莹杉　张婧瑶

Tongji University

2022.06.09

Table of Contents

1 玻色-爱因斯坦凝聚概述
- BEC 理论及实验背景
- BEC 的物理图像

2 Gross-Pitaevskii 方程
- 理论推导
- 时间劈裂赝谱虚时演化算法
- 结果分析

3 总结与讨论

4 附录

BEC 及实验背景

Bose–Einstein condensate[1]：理想玻色气体温度降低到对应的临界温度之下后，宏观数量的玻色子将占据同一个量子态，发生凝聚现象。BEC 是典型的**宏观量子态**，具有高度可控的量子特性，其制备是量子模拟和量子精密测量研究的基础。

- Einstein 预言 BEC 的存在 (**1925**)→ London 提出超流态液氦本质上是 BEC(1938) → 实验上观察到 ^{87}Rb 的 BEC 现象 (**1995**) [2]
- 在“原子气体玻色爱因斯坦凝聚体虚拟仿真实验”中，运用了激光冷却、磁光陷阱冷却和蒸发冷却等技术，制备了 ^{87}Rb 的 BEC。由飞行时间图像，处理得到温度在 100 nK 左右。



图 6 蒸发冷却获得 BEC

[1] https://en.wikipedia.org/wiki/Bose-Einstein_condensate
[2] 吴遒. 玻色-爱因斯坦凝聚体中的基态和涡旋环动力学性质研究 [D]. 扬州大学,2020.DOI:10.27441

BEC 系统的统计理论

考虑由 N 个全同、近独立的玻色子组成的系统，温度为 T，体积为 V，粒子数密度 $n=N/V$。根据玻色分布，定义化学势 $\mu<0$ 满足[3]：

$$\because \frac{2\pi}{h^3}(2m)^{3/2}\int_0^\infty \frac{\varepsilon^{1/2}\mathrm{d}\varepsilon}{\exp(\frac{\varepsilon-\mu}{kT})-1}=n=\mathrm{constant}\qquad \therefore T\to T_C,\qquad \mu\to -0.$$

$$\text{critical temperature}:\quad T_C=\frac{2\pi}{(2.612)^{2/3}}\frac{\hbar}{mk}(n)^{2/3}$$

$$n_0(T)+\frac{2\pi}{h^3}(2m)^{3/2}\int_0^\infty \frac{\varepsilon^{1/2}\mathrm{d}\varepsilon}{\exp(\frac{\varepsilon}{kT})-1}=n,\qquad (T\le T_C)$$

处在最低能级的粒子数密度为 $n_0(T)=n\left[1-\left(\frac{T}{T_C}\right)^{3/2}\right]$

温度低于 T_C 时，温度越低，凝聚的原子数比例越高



[3]汪志诚. 热力学统计物理 [M]. 北京：高等教育出版社，2013:230-235

BEC 的物理图像



图 1: 开放耗散的 BEC 系统的物理图像 [4]

[4] http://www.fuw.edu.pl/wiki/images/6/67/TheoreticalDescriptionofBEC_NB.pdf

GP 方程的理论推导

Gross–Pitaevskii equation[5]：使用 Hartree-Fock 近似和赝势模型描述了由全同玻色子组成的量子多体系统的基态。

$$\Psi(\mathbf{r}_1, \mathbf{r}_2, \cdots, \mathbf{r}_N) = \psi(\mathbf{r}_1)\psi(\mathbf{r}_2)\cdots\psi(\mathbf{r}_N)$$

$$\mathcal{H} = \sum_{i=1}^{N}\left[-\frac{\hbar^2}{2m}\nabla_i^2 + V(\mathbf{r}_i)\right] + g\sum_{i<j}\delta(\mathbf{r}_i - \mathbf{r}_j), \quad \text{where } g = \frac{4\pi\hbar^2 a_s}{m}$$

重新定义一个**凝聚体波函数** $\Psi(\mathbf{r}) = \sqrt{N}\psi(\mathbf{r})$，满足归一化条件 $\int |\Psi(\mathbf{r})|^2 \mathrm{d}r^3 = N$

Gross–Pitaevskii equation

$$\left[-\frac{\hbar^2}{2m}\nabla^2 + V(\mathbf{r}) + g|\Psi(\mathbf{r})|^2\right]\Psi(\mathbf{r}) = \mu\Psi(\mathbf{r})$$

$$i\hbar\frac{\partial}{\partial t}\Psi(\mathbf{r}, t) = \left[-\frac{\hbar^2}{2m}\nabla^2 + V(\mathbf{r}) + g\left|\Psi(\mathbf{r}, t)\right|^2\right]\Psi(\mathbf{r}, t)$$

[5] https://en.wikipedia.org/wiki/Gross-Pitaevskii_equation

时间劈裂赝谱虚时演化算法

$$i\hbar\frac{\partial}{\partial t}\psi(x,t)=\mathbf{A}\psi(x,t),\quad \mathbf{A}=T+V+G=-\frac{\hbar^2}{2m}\frac{\mathrm{d}^2}{\mathrm{d}x^2}+V(x)+gN|\psi(x,t)|^2$$

Time-splitting spectral：通过傅里叶变换，**使动能和势能在各自对角的表象下发挥作用**

$$U(\Delta t)\approx \mathrm{e}^{-i\frac{V+G}{2}\Delta t}\mathrm{e}^{-iT\Delta t}\mathrm{e}^{-i\frac{V+G}{2}\Delta t}$$

TSSP 的数值离散格式

$$\psi_j := \psi_j\cdot\exp\left[-\frac{i\Delta t}{2}(V(x_j)+G(\psi_j))\right]$$

$$\phi \Leftarrow \mathrm{FFT} \Leftarrow \psi$$

$$\phi_j := \phi_j\cdot\exp\left(-i\Delta t\cdot\frac{k_j^2}{2m}\right)$$

$$\psi \Leftarrow \mathrm{IFFT} \Leftarrow \phi$$

$$\psi_j := \psi_j\cdot\exp\left[-\frac{i\Delta t}{2}(V(x_j)+G(\psi_j))\right]$$

时间劈裂賡谱虚时演化算法

$$i\hbar\frac{\partial}{\partial t}\psi(x,t)=\mathbf{A}\psi(x,t),\quad \mathbf{A}=T+V+G=-\frac{\hbar^2}{2m}\frac{\mathrm{d}^2}{\mathrm{d}x^2}+V(x)+gN|\psi(x,t)|^2$$

Imaginary-time propagation method：对于含时薛定谔方程，令**虚时间** $\tau=it$，演化最终得到系统基态

虚时演化原理

$$\psi(x,\tau)=\sum_i c_i\mathrm{e}^{-E_i\tau}\psi_i(x)=\mathrm{e}^{-E_0\tau}\left[c_0\psi_0+\sum_{i=1}c_i\mathrm{e}^{-(E_i-E_0)\tau}\psi_i(x)\right]$$

$$\psi(x,\tau)\ \xrightarrow{\text{long }\tau}\ \psi_0(x)\mathrm{e}^{-E_0\tau}$$

数值方法分析：

- Krylov 迭代的问题：本征矢不正交，容易收敛到不需要的本征值。
- 空间有限差分的缺点：空间精度比较低，同时配合使用显式龙格库塔方法则不稳定，高频噪声会迅速发散。
- TSSP 的优点：具有很高的精度的同时数值稳定性非常强。

结果分析

GP 方程中相互作用的耦合项对 BEC 系统特性的影响

表 1: 谐振子势参数与耦合系数

ω	$g_{\text{non}}N$	$g_{\text{rep}}N$	$g_{\text{att}}N$
1.0	0.0	5.0	−2.0

表 2: 系统化学势 μ（数值结果）

μ_{non}	μ_{rep}	μ_{att}
0.50	2.02	−0.48



图 2: 展示了在谐振子外部势场下，玻色子间相互作用分别为无/排斥/吸引的 BEC 系统的凝聚体波函数基态。排斥相互作用使波函数更弥散，吸引相互作用使波函数更集中。

结果分析

特别地，一维凝聚体波函数的基态可能会出现简并

$$V(x) = \frac{1}{2}x^2 + \frac{1}{20}x^4 + 4\exp(-2x^2), \quad gN = -0.5$$



(a) 收敛到基态波函数其一



(b) 收敛到基态波函数其二

图 3: 在对称初态上施加不同的初始噪声，最后收敛的基态波函数，左右分别为两个随机可能出现的情况。可以看出在一些情况下，吸引势造成了对称破缺，凝聚体波函数基态不再具有哈密顿量的空间反演对称性，能量基态出现二重简并。

总结与讨论

1. 对 **BEC** 理论和实验背景，及量子力学对 BEC 系统的微观描述—**GP 方程**进行了探究
2. 使用**时间劈裂赝谱方法**结合**虚时演化方法**数值求解了含时 GP 方程
3. 定性讨论了 GP 方程中相互作用的耦合项对 BEC 系统特性的影响
4. 观察到“一维凝聚体波函数的基态可能会出现简并”的特殊现象
5. 对虚时演化和实际实验中的动力学过程的关联进行了简单讨论

小组分工

- 张婧瑶　学号：1952498　负责 BEC 与 GP 方程概述部分
- 赵莹杉　学号：1952275　负责数值求解 GP 方程与分析部分

代码附录

```julia
## TSSP Method for 1D Non-linear Schrödinger Equation
# ψ0, x should be 1D arrays of the same size.
# V(x,ψ) is the non-linear potential.
using LinearAlgebra
using FFTW

@inline function iV_step!(ψn, V, x, dt, Nx)
    @inbounds for i in 1:Nx
        ψn[i] *= exp(-V(x[i],ψn[i])*dt)
    end
end

@inline function iT_step!(ψn, ω_g, dt, Nx)
    fft!(ψn)
    @inbounds for i in 1:Nx
        ψn[i] *= exp(-ω_g[i]*dt)
    end
    ifft!(ψn)
end

function iTSSP(ψ0, V, x, t)
    dx = x[2] - x[1]
    Nx = length(x); Nt = length(t)
    # k空间的离散点
    k_g = (2π / dx) .* (((-Nx÷2)//Nx) : (1//Nx) : ((Nx÷2-1)//Nx))
    # 计算对应频率并且按照fft的顺序重排
    # ω_g = (k_g.^2/2)
    ω_g = circshift((k_g.^2/2) , -(Nx÷2))
    ψs = zeros(ComplexF64,Nx,Nt) # 分配数组
    ψs[:,1] .= ψ0 # 初值
    for i in 1:Nt-1
        normalize!(@view ψs[:,i])
        ψs[:,i] ./= √dx
        dt = t[i+1]-t[i]
        ψn = @view ψs[:,i+1]
        copy!(ψn,ψs[:,i])
        # Step V-T-V
        iV_step!(ψn, V, x, dt/2, Nx)
        iT_step!(ψn, ω_g, dt, Nx)
        iV_step!(ψn, V, x, dt/2, Nx)
    end
    return ψs
end
```

```julia
using BenchmarkTools
using Plots, LaTeXStrings, Measures; gr()
include("TSSP1D.jl")

xgrid = LinRange(-6,6,800)
tgrid = LinRange(0,1000,50000)
C = 1/24/(xgrid[2]-xgrid[1])^2
function energy(q,Vfunc)
    V = Vfunc.(xgrid,q)
    Aq = C * ( circshift(q,2) - 16*circshift(q,1) + 30*q - 16*circshift(q,-1) + circshift(q,-2) ) + V .* q
    E = real( dot(q,Aq) / dot(q,q) )
end

V(x) = x^2 / 2# + x^4/20 + 4exp(-2x^2)
V0(x,ψ) = V(x)
V1(x,ψ) = V(x) + 5.0 * abs2(ψ)
V2(x,ψ) = V(x) - 2.0 * abs2(ψ)
ψ0 = ComplexF64[ exp(-x^2 /2) / (π)^(1/4) + 0.1randn() for x in xgrid]

ψs0 = iTSSP(ψ0, V0, xgrid, tgrid)
ψs1 = iTSSP(ψ0, V1, xgrid, tgrid)
ψs2 = iTSSP(ψ0, V2, xgrid, tgrid)
E0 = energy( ψs0[:,end-1], V0 )
E1 = energy( ψs1[:,end-1], V1 )
E2 = energy( ψs2[:,end-1], V2 )

p1= plot(size = (800,500), minorgrid = true,
        titlefontsize=16,legendfontsize=12,
        guidefontsize=14,tickfontsize=10,
        legend=:topright,margins=5mm)
    title!("ground-state single-particle wavefunction in BEC")
    plot!(xgrid, x->0.05*(V(x).-V(0)), line=(2,0.5), labels="external potential")
    plot!(xgrid, abs2.(ψs0[:,end-1]), line=(4,0.5,:dot), labels="non-interaction")
    plot!(xgrid, abs2.(ψs1[:,end-1]), line=(3,0.5), labels="repulsive interaction")
    plot!(xgrid, abs2.(ψs2[:,end-1]), line=(3,0.5), labels="attractive interaction")
    xaxis!(L"x")
    yaxis!(L"|\psi|^2")
    xlims!(-5,5)
    ylims!(-0.2,0.8)
savefig(p1,"GPE.svg")

p2= plot(size = (800,500), minorgrid = true,
        titlefontsize=16,legendfontsize=12,
        guidefontsize=14,tickfontsize=10,
        legend=:topleft,margins=5mm)
    title!("Degenerate ground state")
    plot!(xgrid, x->0.05*(V(x).-V(0)), line=(2,0.5), labels="external potential")
    plot!(xgrid, abs2.(ψs0[:,end-1]), line=(3,0.5,:dot), labels="non-interaction")
    plot!(xgrid, abs2.(ψs2[:,end-1]), line=(3,0.5), labels="attractive interaction")
    xaxis!(L"x")
    yaxis!(L"|\psi|^2")
    xlims!(-5,5)
    ylims!(-0.2,0.8)
```